
ABSTRACT 
When dealing with large amounts of longitudinal data, it is often necessary to divide data pertaining to the 
same individual among multiple datasets, representing data from different points of time.  Combining data 
from such datasets will be particularly difficult if these different datasets do not contain both a unique 
identifying index variable (e.g., IDNUM) and a temporal index variable (e.g., Visit_date). The current paper 
describes a public health analysis in which PROC SQL and subsequent DATA steps are used to link 
temporally related data relatively easily. 

INTRODUCTION 
As many data managers will agree, the sheer amount of data one may collect when doing longitudinal 
analyses often necessitates the need to place data pertaining to different variables into different datasets.  
Moreover, it is often the case that the data in these different datasets, though pertaining to the same 
individual, relate to different points in time (that is, data in one dataset may be collected around the time of 
data in another dataset, without being collected on the same date).  While these different datasets will often 
contain an identical index variable (e.g., IDNUM), if they do not also have an identical temporal index 
variable, a typical many-to-many merge is likely an inefficient method of combining the data.  Fortunately, 
one can create a table using the PROC SQL that makes combining these data relatively easy. 

AN EXAMPLE 
Longitudinal cohort studies are very common in public health, particularly in areas of public health 
investigating HIV infection and AIDS.  In a typical cohort study, participants will provide personal and 
demographic data, HIV risk data, psychosocial data, and data relating to various biomedical tests.  
Depending on the size of the cohort and the length of the study, these datasets can become huge and each 
type of data may be placed into different datasets.   

For the purposes of this paper, I am going to discuss data from a longitudinal cohort study of HIV infected 
persons contained in two different datasets.  The first dataset, VISITS, contains personal information about a 
study participant (e.g., age, height, and weight) at a specific point in time (identified by the temporal variable 
VX_DATE).  The second dataset, ALLVB, contains data regarding the results of a lab test to detect the 
presence of HIV antibodies in a participant’s blood at a specific point in time (identified by the temporal 
variable, VBDATE).  Unfortunately, while VX_DATE and VBDATE can sometimes be the same, physicians 
in this study often order lab tests either before or after a particular office visit. Thus, VBDATE may actually 
precede or follow VX_DATE.  As if this were not challenging enough, the data in each of these datasets 
corresponds to information collected when the study participant goes to see the doctor.  Because a study 
coordinator did not schedule visits, there is no consistency among participants regarding the total number of 
visits each participant has.  That is, some participants may have weekly visits; some may have monthly 
visits, while others may only have visits when they are sick.  Finally, sometimes a participant can come in 
for a doctor’s visit without any labs being drawn; thus, participants may have some visits without 
corresponding lab data.  

A simple many-to-many merge would only work in the instances in which VBDATE and VX_DATE were 
identical (providing one used some kind of renaming system so that both datasets would have date variables 
with the same name).  If one used this method, data from those individuals with different date variables 
would be lost.  I will demonstrate in this paper a heuristic to link the data in the VISITS and ALLVB 
datasets, even if the date variables in the datasets are not identical. 

APPLICATION AND PROGRAMMING 
We start with two datasets.  The first, VISIT, contains several variables including HRNID (study id number) 
and VX_DATE (visit date).  The second data set, ALLVLOAD, contains several variables including HRNID, 
VBDATE (date of viral load assessment), and VLOAD (result of viral load assessment).  Looking at the SAS 
code below, the first line of the PROC SQL statement defines a new SAS dataset (table) called VISIT_VL.  
The SELECT clause defines which variables to include in the table and the FROM option defines the 
datasets that will be used to create the table.  The WHERE option states that the only condition of this 
procedure is that the ID variables from each dataset have to be the same.  Thus, this PROC SQL statement 
will create a dataset in which every observation for a given HRNID in the VISIT dataset is paired with every 
observation for that HRNID in the ALLVLOAD dataset. 
 
proc sql; 
 create table visit_vl as 
 select visit.hrnid, vx_date, vbdate, vload  

From visit, allvload  
  where visit.hrnid = allvload.hrnid; 
run; 
 
In the study that provided these data, it is possible that viral load information may be obtained before, during, 
or after a patients visit.  However, we are only interested in retaining the viral load information if it was 
obtained within 30 days of a given visit.  Thus, the next piece of code takes this newly created table, 
computes the absolute difference between VX_DATE and VBDATE and deletes any observation in which 
this difference is greater than 30 days.   
 
data timediff; 
 set visit_vl ; 
 absdiff = ABS(vx_date - vbdate) ; 
 if absdiff > 30 then delete ; 
RUN; 
 
The next step in this process is to sort the dataset by ID, visit date, and the absolute difference between 
visit date and the date of viral load assessment.  The BY command used below causes SAS to create the 
FIRST. logical variable (it also creates the LAST. logical variable, but we do not use it in this paper).  When 
SAS encounters the first observation within the BY group (in the code below, this would be the first 
observation of each particular VX_DATE for every HRNID), FIRST.VX_DATE is true; otherwise it is false.  
Once the data are sorted in this manner, using the FIRST. command in a data step allows us to select the 
viral load assessment that occurred closest in time to each specific visit for a given participant (that is, the 
observation with the smallest value for ABSDIFF for each visit a participant had). 
 
proc sort data=timediff ; 
 by hrnid vx_date absdiff; 
run; 
 
data data1 ; 
 set timediff ; 
 by hrnid vx_date ; 
 if first.vx_date ; 
run; 
 
There is one final step in this process.  It is possible that a particular individual might have multiple visits 
within a relatively short time frame, yet only have his or her viral load assessed once.  In this instance, the 
above code may end up pairing the same viral load data with multiple visits.  To address this problem, the 
dataset is again sorted, but this time by ID, date of viral load assessment, and the absolute difference 
between visit date and the date of viral load assessment.  Once the data are sorted in this manner, using 
the FIRST. command allows us to select the closest visit to each viral load assessment. 

proc sort data=data1 ; 
 by hrnid vbdate absdiff; 
run ; 
 
data final ; 
 set data1 ; 
 by hrnid vbdate ; 
 if first.vbdate ; 
 drop absdiff ; 
run; 

Using this procedure, the FINAL dataset will contain, for each study participant, all of the visit dates for 
which there was unique viral load data; the many to many merge is complete.  Here is the complete code 
(this is the same as above, but without the commentary separating the sections). 
 
proc sql; 
 create table visit_vl as 
 select visit.hrnid, vx_date, vbdate, vload From visit, allvload  
  where visit.hrnid = allvload.hrnid; 
run; 
 
 
data timediff; 
 set visit_vl ; 
 absdiff = ABS(vx_date - vbdate) ; 
 if absdiff > 30 then delete ; 
run; 
 
 
proc sort data=timediff ; 
 by hrnid vx_date absdiff; 
run; 
 
 
data data1 ; 
 set timediff ; 
 by hrnid vx_date ; 
 if first.vx_date ; 
run; 
 
 
proc sort data=data1 ; 
 by hrnid vbdate absdiff; 
run ; 
 
 
data final ; 
 set data1 ; 
 by hrnid vbdate ; 
 if first.vbdate ; 
 drop absdiff ; 
run; 

CONCLUSION 
This code, though used here within a public health example, has applications across disciplines that deal 
with longitudinal data.  This methodology should help you if you ever need to combine longitudinal data that 
may be temporally associated, but perhaps not collected at exactly the same time. 

